Supplementary Information for:

Mitotic trigger waves and the spatial coordination of the *Xenopus* cell cycle

Jeremy B. Chang[1], James E. Ferrell, Jr.[1,2*]

[1]Department of Chemical and Systems Biology, Stanford University School of Medicine, Stanford CA 94305-5174, USA.

[2]Department of Biochemistry, Stanford University School of Medicine, Stanford CA 94305-5307.

Computational Modeling

To model mitotic trigger waves in *Xenopus* egg extracts, we began with two ordinary differential equations that describe the synthesis/formation and degradation of cyclin B1-Cdk1 complexes and the interconversion of cyclin B1-Cdk1 between active and inactive phosphorylation states, and then added diffusion terms. The Hill functions in the ODEs were taken from ref 1 and are meant to encapsulate all of the pathways through which active Cdk1 regulates Cdc25, Wee1, and APC/C, including multisite phosphorylation, regulation of Greatwall and PP2A-B55, and other as yet unknown pathways. The parameters for the ODE models were taken from ref 1. The resulting PDEs were:

$$\frac{\partial Cdk1_{act}}{\partial t} = Dp\nabla^2 Cdk1_{act} + k_{synth} + \left(a_{Cdc25} + b_{Cdc25}\frac{Cdk1_{act}^{n_{Cdc25}}}{EC50_{Cdc25}^{n_{Cdc25}} + Cdk1_{act}^{n_{Cdc25}}}\right)Cdk1_{inact} -$$
$$\left(a_{Wee1} + b_{Wee1}\frac{K_{Wee1}^{n_{Wee1}}}{EC50_{Wee1}^{n_{Wee1}} + Cdk1_{act}^{n_{Wee1}}}\right)Cdk1_{act} - \left(a_{deg} + b_{deg}\frac{Cdk1_{act}^{n_{deg}}}{EC50_{deg}^{n_{deg}} + Cdk1_{act}^{n_{deg}}}\right)Cdk1_{act}$$

$$\frac{\partial Cdk1_{inact}}{\partial t} = Dp\nabla^2 Cdk1_{inact} - \left(a_{Cdc25} + b_{Cdc25}\frac{Cdk1_{act}^{n_{Cdc25}}}{EC50_{Cdc25}^{n_{Cdc25}} + Cdk1_{act}^{n_{Cdc25}}}\right)Cdk1_{inact}$$
$$+ \left(a_{Wee1} + b_{Wee1}\frac{K_{Wee1}^{n_{Wee1}}}{EC50_{Wee1}^{n_{Wee1}} + Cdk1_{act}^{n_{Wee1}}}\right)Cdk1_{act} - \left(a_{deg} + b_{deg}\frac{Cdk1_{act}^{n_{deg}}}{EC50_{deg}^{n_{deg}} + Cdk1_{act}^{n_{deg}}}\right)Cdk1_{inact}$$

where $Cdk1_{inact}$ denotes inactive cyclin B1-Cdk1 complexes and $Cdk1_{act}$ denotes active complexes. The parameters were:

$$a_{cdc25} = 0.8 \text{ min}^{-1}$$
$$b_{cdc25} = 4 \text{ min}^{-1}$$
$$n_{cdc25} = 11$$
$$EC50_{cdc25} = 35 \text{ nM}$$
$$a_{wee1} = 0.4 \text{ min}^{-1}$$
$$b_{wee1} = 2 \text{ min}^{-1}$$
$$EC50_{wee1} = 30 \text{ nM}$$
$$n_{wee1} = 3.5$$
$$a_{deg} = 0.01 \text{ min}^{-1}$$
$$b_{deg} = 0.06 \text{ min}^{-1}$$
$$EC50_{deg} = 32 \text{ nM}$$
$$n_{deg} = 17$$

The diffusion constant for both species, *Dp*, was taken to be 600 μm^2/min (10 μm^2/sec). The Mathematica code for numerically solving the two PDEs was:

```
scale = 10;
EC50deg = 32;
EC50cdc25 = 35;
EC50wee1 = 30;
tmax = 300;
Dp = 600;
xrange = 2000;

sol = NDSolve[
  {
  D[Cdk1[t, x], t] ==
   +ksynth[t, x]
   + (acdc25[t, x] +
     bcdc25[t, x] Cdk1[t, x]^11/(
     EC50cdc25^11 + Cdk1[t, x]^11)) (Cdk1inact[t, x])
   - (awee1[t, x] +
     bwee1[t, x] EC50wee1^3.5/(EC50wee1^3.5 + Cdk1[t, x]^3.5)) Cdk1[t, x]
   - (adeg[t, x] +
     bdeg[t, x] Cdk1[t, x]^17/(EC50deg^17 + Cdk1[t, x]^17)) Cdk1[t, x]
   + Dp D[Cdk1[t, x], x, x],
  D[Cdk1inact[t, x], t] ==
   - (adeg[t, x] +
     bdeg[t, x] Cdk1[t, x]^17/(EC50deg^17 + Cdk1[t, x]^17)) (Cdk1inact[t,
     x])
   - (acdc25[t, x] +
     bcdc25[t, x] Cdk1[t, x]^11/(
     EC50cdc25^11 + Cdk1[t, x]^11)) (Cdk1inact[t, x])
   + (awee1[t, x] +
     bwee1[t, x] EC50wee1^3.5/(EC50wee1^3.5 + Cdk1[t, x]^3.5)) Cdk1[t, x]
   + Dp D[Cdk1inact[t, x], x, x],

  D[adeg[t, x], t] == 0,
  D[bdeg[t, x], t] == 0,
  D[acdc25[t, x], t] == 0,
```

```
  D[bcdc25[t, x], t] == 0,
  D[awee1[t, x], t] == 0,
  D[bwee1[t, x], t] == 0,
  D[ksynth[t, x], t] == 0,

  ksynth[0, x] == 1.5(*2.5*),
  adeg[0, x] == .01,
  bdeg[0, x] == .06,
  awee1[0, x] == scale*.08*.5,
  bwee1[0, x] == scale*.4*.5,
  acdc25[0, x] == scale*Piecewise[
    {
    {0.08, x < -2.5},
    {.12, -2.5 <= x < 2.5},
    {0.08, 2.5 <= x}
    }
    ],
  bcdc25[0, x] == scale*Piecewise[
    {
    {0.4, x < -2.5},
    {.6, -2.5 <= x < 2.5},
    {0.4, 2.5 <= x}
    }
    ],
  Cdk1inact[0, x] == 0,
  Cdk1[0, x] == 0,

  (D[Cdk1[t, x], x] /. x -> -xrange) == 0,
  (D[Cdk1[t, x], x] /. x -> xrange) == 0,
  (D[Cdk1inact[t, x], x] /. x -> -xrange) == 0,
  (D[Cdk1inact[t, x], x] /. x -> xrange) == 0

  },
  {Cdk1, Cdk1inact, acdc25, bcdc25, awee1, bwee1, ksynth, adeg, bdeg},
  {t, 0, tmax},
  {x, -xrange, xrange},
  MaxStepFraction -> .005,
  MaxSteps -> 100000];
```

The expression used to plot the results was:

```
xrangeplot = 2000;
tmaxplot = 300;

contourDensityPlot[Evaluate[Cdk1[t, x] /. sol],
 {t, 0, tmaxplot},
 {x, -xrangeplot, xrangeplot},
 ColorFunction -> Function[{p, q}, ColorData["Rainbow"][.9 p/70 + .1]],
 ColorFunctionScaling -> False,
 PlotPoints -> 300,
 MaxRecursion -> 4,
 PlotRange -> {0, 200},
 AspectRatio -> ((xrangeplot/tmaxplot)/50),
 Contours -> None]
```

The function “contourDensityPlot” is a custom Mathematica function written by Jens Nöckel[2].

Note that the model used here implicitly assumes that there is no time lag between Cdk1 activation and APC/C activation. However, in extracts there is actually about a 15 min time lag

between the two events[3]. We have also carried out simulations with models that account this time lag by assuming that APC/C activation is a multistep process. However, for present purposes, the simple no-time-lag, 2-PDE model suffices.

PDE modeling of a negative-feedback-only oscillator circuit. For comparison, one can also model oscillations of Cdk1 activity due to a negative-feedback-only oscillator circuit. Since a two-variable negative feedback loop will not oscillate without a bistable trigger, we added an intermediary between Cdk1 activation and APC/C activation, assuming that Cdk1 activates Plx1 and Plx1 activates APC/C. We also assumed, as we did in Fig 1d, that Cdc25 activity was 50% higher in the 5 μm middle of the tube, and then asked how the oscillations would evolve in time and space.

The code for the model is:

```
scale = 100;

EC50deg = 32;
EC50cdc25 = 35;
EC50wee1 = 30;
kplx1on = 1.5;
kplx1off = .15;
kapcon = 1.5;
kapcoff = .15;
plx1tot = 1;
nplx1 = 5;
EC50plx1 = 60;
EC50apc = 0.5;
apctot = 1;
napc = 4;
kdest = 1;
tmax = 100;
Dp = 600;
xrange = 666;
bwee1 = scale*.1;

pdesol2 = NDSolve[
   {

    D[Cdk1[t, x], t] ==
     +ksynth[t, x]
      + ((*acdc25[t,x] +*) bcdc25[t, x] ) (Cdk1inact[t, x])
      - (bwee1) Cdk1[t, x]
      - kdest*apc[t, x]*Cdk1[t, x]
      + Dp D[Cdk1[t, x], x, x],
    D[Cdk1inact[t, x], t] ==
     - kdest*apc[t, x]*(Cdk1inact[t, x])
      - ((*acdc25[t,x] +*) bcdc25[t, x] ) (Cdk1inact[t, x])
      + (bwee1) Cdk1[t, x]
      + Dp D[Cdk1inact[t, x], x, x],
    D[plx1a[t, x], t] == -kplx1off*plx1a[t, x]
      + kplx1on*(plx1tot - plx1a[t, x])*(Cdk1[t, x]^nplx1/(
```

```
        EC50plx1^nplx1 + Cdk1[t, x]^nplx1)),
    D[apc[t, x], t] == -kapcoff*apc[t, x]
      + kapcon*(apctot - apc[t, x])*(plx1a[t, x]^napc/(
        EC50apc^napc + plx1a[t, x]^napc)),

    D[bcdc25[t, x], t] == 0,
    D[ksynth[t, x], t] == 0,
    ksynth[0, x] == 1,
    bcdc25[0, x] == scale*Piecewise[
       {
        {0.4, x < -2.5},
        {0.6, -2.5 <= x < 2.5},
        {0.4, 2.5 <= x}
        }
       ],
    Cdk1inact[0, x] == 0,
    Cdk1[0, x] == 0,
    apc[0, x] == 0,
    plx1a[0, x] == 0,

    (D[Cdk1[t, x], x] /. x -> -xrange) == 0,
    (D[Cdk1[t, x], x] /. x -> xrange) == 0,
    (D[Cdk1inact[t, x], x] /. x -> -xrange) == 0,
    (D[Cdk1inact[t, x], x] /. x -> xrange) == 0
    },
   {Cdk1, Cdk1inact, acdc25, bcdc25, ksynth,  plx1a, apc},
   {t, 0, tmax},
   {x, -xrange, xrange},
   MaxStepFraction -> .0002];
Plot[{Evaluate[Cdk1[t, 0] /. pdesol2], Evaluate[Cdk1[t, 500] /. pdesol2],
  Evaluate[Cdk1inact[t, 0] /. pdesol2],
  Evaluate[Cdk1inact[t, 500] /. pdesol2]}, {t, 0, tmax}, PlotRange -> All]
```

The results are shown in Supplementary Figure 1. The character of the oscillations is more sigmoidal and less spiky, so that there are gradual changes in the heat map representation of Cdk1 activation as you go from left to right. The centrosomal region gives rise to a narrow streak of early Cdk1 activation in the middle of the plot, with no significant spread of Cdk1 up and down the tube.

Modeling surface contraction waves. To model whether surface contraction waves could be the direct result of an expanding spherical trigger wave of Cdk1 activity, we solved for the intersection points of two circles, one with radius R_C centered at (0,0) (Supplementary Figure 4, blue), and one with variable radius r centered at (x_0,y_0) (Supplementary Figure 4, red). The radius r can be expressed in terms of t_0, the time at which the trigger wave began, and v, the speed at which the trigger wave is expanding, through the equation $r = v(t - t_0)$. The intersection points (x,y) were solved for using Mathematica with the following code:

```
Solve[y^2 + x^2 == rc^2 && (y – y0)^2 + (x – x0)^2 == v^2*(t-t0)^2, {x, y}]
```

Here we are particularly interested in the solution for x, since that is what we can most easily measure by video microscopy:

$$x = \frac{x_0\left(R_c^2 - \left(t - t_0\right)^2 v^2 + x_0^2 + y_0^2\right) \pm \sqrt{-y_0^2\left(x_0^2 + y_0^2 - \left(R_c + \left(t - t_0\right)v\right)^2\right)\left(x_0^2 + y_0^2 - \left(R_c - \left(t - t_0\right)v\right)^2\right)}}{2\left(x_0^2 + y_0^2\right)}$$

This equation can be rearranged to express t as a function of x and the parameters:

$$t = t_0 + \frac{1}{v}\sqrt{R_c^2 - 2xx_0 + x_0^2 + y_0^2 - 2\sqrt{\left(R_c^2 - x^2\right)y_0^2}} \qquad \text{[Eq S1]}$$

We then used nonlinear regression to fit Eq S1 to the data using:

```
NonlinearModelFit[
 data,
 {(t0-(1/v)*Sqrt[rc^2-2x*x0+x0^2+y0^2-2*Sqrt[(rc^2-x^2)*y0^2]],
 {0 < y0 < rc, -rc < x0 < rc, x0^2 + y0^2 < rc^2, v > 0}},
 {x0, y0, t0, v},
 x,
 MaxIterations -> 2000,
 ConfidenceLevel->.99,
 Method -> "NMinimize"]
```

Supplementary Figures and Figure Legends:



Supplementary Figure 1 | Cdk1 activation as a function of time and position in a one-dimensional tube, assuming that there is no bistable trigger in the oscillator circuit. As was the case in Fig. 1d, it was assumed that cyclin B1 is synthesized at a uniform rate everywhere in the tube, but in a 5 μm region in the middle of the tube the concentration of Cdc25C is 50% higher than in the rest of the tube, allowing Cdk1 to become activated earlier. Cyclin B1-Cdk1 activity is denoted by the color scale (blue is low, red is high). Numerical solution of the PDEs was carried out using Mathematica 9.0 (Wolfram). Note that the peak of Cdk1 activity is lower here than in the cycles shown in Fig. 1d, so the heat map scale was adjusted to make it easier to perceive the oscillations here.



Supplementary Figure 2 | The propagation rate slows down progressively. Wave speeds are shown for mitotic entry (circles) and exit (triangles) as means ± S.D. Data are from one experiment (the experiment shown in Fig. 2B).



Supplementary Figure 3 | Effects of PD0166285 on Cdk1 Y15 phosphorylation and mitotic waves. a, Time course of Cdk1 Y15 phosphorylation. **b**, Dose/response data. Extracts were incubated with 1 mM sodium orthovanadate to inhibit Cdc25C. After 39 min, non-degradable Δ65-cyclin B1 and various concentrations of PD0166285 were added. Samples were taken at 69 min for pY15 immunoblotting. In panels A and B, pY15-Cdk1 signal was quantified with a LI-COR Odyssey Imager. The IC50 was estimated to be 0.74 μM. **c-e**, Effects of PD0166285 on wave speed (**c**), interphase duration (**d**), and mitotic duration (**e**) in various cycles. In panel C the circles correspond to mitotic entrance and the triangles to mitotic exit. Interphase and M-phase duration were calculated from the data in Fig. 3 by defining polygonal interphase and M-phase regions by linear interpolation (the pink and blue regions in Fig. 3), dividing these regions into 10 equal-length segments, and calculating the area divided by the height for each region. Durations are expressed as means ± S.D.



Supplementary Figure 4. | Calculating the time at which a constant velocity, spherical wave of Cdk1 activation would arrive at the cell cortex. The blue circle represents an egg of radius R_c. The red circle represents a spherical trigger wave that originated at a position (x_0,y_0) and has a radius $r = v(t - t_0)$.

Supplemental References:

1. Trunnell, N. B., Poon, A. C., Kim, S. Y. & Ferrell, J. E., Jr. Ultrasensitivity in the regulation of Cdc25C by Cdk1. *Mol Cell* **41**, 263-74 (2011).

2. J. U. Nöckel, *Mathematica Density and Contour Plots with Rasterized Image Representation*, http://pages.uoregon.edu/noeckel/MathematicaGraphics.html.

3. Yang, Q. & Ferrell, J. E., Jr. The Cdk1-APC/C cell cycle oscillator circuit functions as a time-delayed ultrasensitive switch. *Nat Cell Biol* **15**, 519-25, (2013).